
Пример реализации разностной схемы для задачи 

с учетом условий сопряжения: 
 
 
// Задание сетки------------------------------------------------------------ 
L=1;                      // длина расчетного участка 
x0=0.33333333;   // точка разрыва функции k(x) 
N1=11;                 // число узлов по x 
h1=x0/(N1-1);     // шаг по x при x<=x0 
N=N1+round((L-x0)*(N1-1)/x0); 
h2=(L-x0)/(N-N1); 
x=zeros(N,1); 
for n=1:N 
    if n<=N1 
        x(n)=(n-1)*h1; 
    else 
        x(n)=x0+(n-N1)*h2; 
    end 
end 
 
// Входные данные задачи---------------------------------------------------- 
k1=1;          // значение k(x) при x<x0 
k2=2;          // значение k(x) при x>x0 
g1=1; 
g2=0; 
// Аналитическое решение---------------------------------------------------- 
u=zeros(N,1); 
kappa=k1/k2; 
a=1/(kappa+(1-kappa)*x0); 
b=kappa*a; 
for n=1:N 
    if x(n)<=x0 
        u(n)=1-a*x(n); 
    else 
        u(n)=b*(1-x(n)); 
    end 
end 
 
// Численное решение-------------------------------------------------------- 
y=zeros(N,1); 
err=zeros(N,1); 
alpha=zeros(N-1,1); 
betta=zeros(N-1,1); 



betta(1)=g1; 
// Прямой ход прогонки: 
for n=2:N-1 
    if n<N1 
        A=k1/h1^2; 
        B=k1/h1^2; 
        C=2*k1/h1^2; 
    end 
    if n==N1 
        A=k1/h1; 
        B=k2/h2; 
        C=A+B; 
    end 
    if n>N1 
        A=k2/h2^2; 
        B=k2/h2^2; 
        C=2*k2/h2^2; 
    end 
    alpha(n)=B/(C-A*alpha(n-1)); 
    betta(n)=A*betta(n-1)/(C-A*alpha(n-1)); 
end 
// Обратный ход прогонки: 
y(N)=g2; 
for n=N-1:-1:1 
    y(n)=alpha(n)*y(n+1)+betta(n); 
    err(n)=abs(y(n)-u(n)); 
end 
//-------------------------------------------------------------------------- 
figure 
plot(x,u,'linewidth',2) 
xlabel('x') 
ylabel('u') 
title('analitical solution') 
 
figure 
plot(x,y,'linewidth',2) 
xlabel('x') 
ylabel('y') 
title('calculated solution') 
 
figure 
plot(x,err*100/max(u),'linewidth',2) 
xlabel('x') 
ylabel('err, %') 
title('error') 
 

 

 

 



Пример реализации консервативной разностной схемы для задачи 

полученной методом баланса: 
 

// Задание сетки------------------------------------------------------------ 
L=1;                    // длина расчетного участка 
x0=0.33333333; // точка разрыва функции k(x) 
N=101;               // число узлов по x 
h=L/(N-1);         // шаг по x  
x=zeros(N,1); 
for n=1:N 
    x(n)=(n-1)*h; 
end 
 
// Входные данные задачи---------------------------------------------------- 
k1=1;          // значение k(x) при x<x0 
k2=2;          // значение k(x) при x>x0 
g1=1; 
g2=0; 
// Аналитическое решение---------------------------------------------------- 
u=zeros(N,1); 
kappa=k1/k2; 
a=1/(kappa+(1-kappa)*x0); 
b=kappa*a; 
for n=1:N 
    if x(n)<=x0 
        u(n)=1-a*x(n); 
    else 
        u(n)=b*(1-x(n)); 
    end 
end 
 
// Численное решение-------------------------------------------------------- 
y=zeros(N,1); 
err=zeros(N,1); 
alpha=zeros(N-1,1); 
betta=zeros(N-1,1); 
betta(1)=g1; 
// Прямой ход прогонки: 
for n=2:N-1 
    if x(n+1)<=x0 
        a=k1; 
        b=k1; 



    end 
    if x(n-1)>=x0 
        a=k2; 
        b=k2; 
    end 
    if (x(n)<=x0)&(x(n+1)>=x0) 
        a=k1; 
        // первый вариант, если b получаем интегрированием: 
         b=h/((x0-x(n))/k1+(x(n+1)-x0)/k2); 
        // второй вариант, если под интегралом берем k(x_n+0.5h) 
//         if (x(n)+x(n+1))/2<=x0 
//             b=k1; 
//         else 
//             b=k2; 
//         end 
    end 
    if (x(n-1)<=x0)&(x(n)>=x0) 
       // первый вариант, если a получаем интегрированием:  
        a=h/((x0-x(n-1))/k1+(x(n)-x0)/k2); 
       // второй вариант, если под интегралом берем k(x_n+0.5h) 
//        if (x(n-1)+x(n))/2<=x0 
//             a=k1; 
//        else 
//             a=k2; 
//        end 
        
       b=k2; 
    end 
    A=a/h^2; 
    B=b/h^2; 
    C=A+B; 
    alpha(n)=B/(C-A*alpha(n-1)); 
    betta(n)=A*betta(n-1)/(C-A*alpha(n-1)); 
end 
// Обратный ход прогонки: 
y(N)=g2; 
for n=N-1:-1:1 
    y(n)=alpha(n)*y(n+1)+betta(n); 
    err(n)=abs(y(n)-u(n)); 
end 
//-------------------------------------------------------------------------- 
figure 
plot(x,u,'linewidth',2) 
xlabel('x') 
ylabel('u') 
title('analitical solution') 
 
figure 
plot(x,y,'linewidth',2) 
xlabel('x') 
ylabel('y') 
title('calculated solution') 



 
figure 
plot(x,err*100/max(u),'linewidth',2) 
xlabel('x') 
ylabel('err, %') 
title('error') 
 


